// Upgrade NOTE: replaced '_Object2World' with 'unity_ObjectToWorld'

Shader "UShaderBook/C7/MormalMapWorldSpace"
{
    Properties
    {
        _Color ("Color Tint", Color) = (1,1,1,1) // 主色调
        _MainTex ("Main Tex", 2D) = "white" {} // 主材质
        _Gloss ("Gloss", Range(8.0,256)) = 20 // 反光度
        _Specular ("Specular", Color) = (1,1,1,1) // 反光颜色
        _BumpMap ("Normal Map", 2D) = "bump" {} // 法线贴图
        _BumpScale ("Bump Scale", Float) = 1.0 // 凹凸程度
    }

    SubShader
    {
        Pass{
            Tags {"LightMode"="ForwardBase"}
            CGPROGRAM
            #pragma vertex vert
            #pragma fragment frag
            #include "Lighting.cginc"
            
            fixed4 _Color;
            sampler2D _MainTex;
            float4 _MainTex_ST;
            float _Gloss;
            fixed4 _Specular;
            sampler2D _BumpMap;
            float4 _BumpMap_ST;
            float _BumpScale;
            
            struct a2v{
                float4 vertex : POSITION;
                float3 normal : NORMAL;
                float4 tangent : TANGENT;
                float4 texcoord : TEXCOORD0;
            };
            
            struct v2f{
                float4 pos : SV_POSITION;
                float4 uv : TEXCOORD0;
                float4 T2W0 : TEXCOORD1; // 世界空间的坐标分散存储在w分量中
                float4 T2W1 : TEXCOORD2;
                float4 T2W2 : TEXCOORD3;
            };
            
            v2f vert(a2v v){
                v2f o;
                // 裁剪空间
                o.pos = UnityObjectToClipPos(v.vertex);
                
                // 纹理坐标
                o.uv.xy = TRANSFORM_TEX(v.texcoord, _MainTex);
                o.uv.zw = TRANSFORM_TEX(v.texcoord, _BumpMap);

                float3 worldPos = mul(unity_ObjectToWorld, v.vertex).xyz; // 世界空间的顶点
                fixed3 worldNormal = UnityObjectToWorldNormal(v.normal);
                fixed3 worldTangent = UnityObjectToWorldDir(v.tangent.xyz);
                fixed3 worldBinormal = cross(worldNormal, worldTangent); // 叉乘是右手系，因此副法线不再是切线乘以法线
                // 法线坐标到世界坐标(法线贴图的制作使用切线、副法线、法线做xyz轴)
                o.T2W0 = float4(worldTangent.x, worldBinormal.x, worldNormal.x, worldPos.x);
                o.T2W1 = float4(worldTangent.y, worldBinormal.y, worldNormal.y, worldPos.y);
                o.T2W2 = float4(worldTangent.z, worldBinormal.z, worldNormal.z, worldPos.z);
                return o;
            }

            fixed4 frag(v2f i) : SV_TARGET{
                float3 worldPos = float3(i.T2W0.w, i.T2W1.w,i.T2W2.w);

                fixed3 lightDir = normalize(UnityWorldSpaceLightDir(worldPos));
                fixed3 viewDir = normalize(UnityWorldSpaceViewDir(worldPos));

                fixed3 bump = UnpackNormal(tex2D(_BumpMap, i.uv.zw));

                bump.xy *= _BumpScale; // 凹凸程度
                bump.z = sqrt(1.0 - saturate(dot(bump.xy, bump.xy)));

                bump = normalize(half3(dot(i.T2W0.xyz, bump), dot(i.T2W1.xyz, bump), dot(i.T2W2.xyz, bump))); // 世界法线
                fixed3 albedo = tex2D(_MainTex, i.uv.xy).rgb * _Color.rgb; // 世界法线
                
                // 环境光反射结果
                fixed3 ambient = UNITY_LIGHTMODEL_AMBIENT.xyz * albedo; 

                // 漫反射结果
                fixed3 diffuse = _LightColor0.rgb * albedo * max(0, dot(bump, lightDir));

                // 高光反射结果
                fixed3 halfDir = normalize(lightDir + viewDir);
                fixed3 specular = _LightColor0.rgb * _Specular.rgb * pow(max(0, dot(bump, halfDir)), _Gloss);

                // return fixed4(i.viewDir, 1);
                return fixed4(ambient + diffuse + specular, 1.0);
            }
            ENDCG
        }
    }
    FallBack "Specular"
}
